data BP;

  input subject visit @7 vsdt mmddyy10. @18 tmtstdt mmddyy10. sysbp diabp @;

  datalines;

1  1  09/15/2007 10/15/2007 140 90 

1  3  10/15/2007 10/15/2007 138 94

1  4  10/29/2007 10/15/2007 142 86 

1  5  11/13/2007 10/15/2007 130 80

1  6  11/29/2007 10/15/2007 132 84 

1  7  12/15/2007 10/15/2007 130 83

1  8  12/30/2007 10/15/2007 131 81 

1  9  01/13/2008 10/15/2007 128 78

2  1  09/01/2007 09/30/2007 122 70 

2  2  09/15/2007 09/30/2007 125 73

2  3  09/30/2007 09/30/2007 128 80 

2  4  10/14/2007 09/30/2007 130 78

2  5  10/30/2007 09/30/2007 128 82 

2  6  11/13/2007 09/30/2007 126 74

2  7  11/28/2007 09/30/2007 128 76 

2  9  12/29/2007 09/30/2007 124 80

;;;;

run;

proc print; run;

/*-------------------------------------------------------------*/

/* Use last value on or before treatment as baseline

/*-------------------------------------------------------------*/

data Change;

  do until(last.subject);

    set BP;

    by subject visit;

    if vsdt<=tmtstdt then do;

      b_sysbp=sysbp;

      b_diabp=diabp;

    end;

  end;

  do until(last.subject);

    set BP;

    by subject visit;

    if vsdt>tmtstdt then do;

      c_sysbp=sysbp-b_sysbp;

      c_diabp=diabp-b_diabp;

    end;

    output;

  end;

run;

proc print noobs;

  format vsdt tmtstdt mmddyy10.;

  by subject;

  id subject;

run;

/*------------------------------------------------------------------------*/

/* Use mean value of all readings on or before treatment as baseline

/*------------------------------------------------------------------------*/

data Change;

  do until(last.subject);

    set BP;

    by subject visit;

    if first.subject then denom=0;

    if vsdt<=tmtstdt then do;

      denom+1;

      s_sysbp=sum(s_sysbp,sysbp);

      s_diabp=sum(s_diabp,diabp);

    end;

  end;

  b_sysbp=s_sysbp/denom;

  b_diabp=s_diabp/denom;

  do until(last.subject);

    set BP;

    by subject visit;

    if vsdt>tmtstdt then do;

      c_sysbp=sysbp-b_sysbp;

      c_diabp=diabp-b_diabp;

    end;

    output;

  end;

run;

proc print noobs;

  format vsdt tmtstdt mmddyy10.;

  by subject;

  id subject;

run;
